# Applied Bayesian Statistics in Animal & Biological Sciences

## 4) Non-linear regression and heteroscedasticity

Henk J. van Lingen

*hvanling@uoguelph.ca*

May 13, 2026

# Today's program

## Methodology

- Bayesian non-linear regression
- Heteroscedasticity and model selection
- Posterior predictive distribution

# Today's program

## Methodology

- Bayesian non-linear regression
- Heteroscedasticity and model selection
- Posterior predictive distribution

## Toxicokinetic modeling datasets

- Zinc and manganese in *Eisenia fetida* (earthworms) exposed to nanomaterials
- Propanolol in *Gammarus Pulex* (Freshwater shrimp/amphipod crustaceans) exposed to polluted water

# General and Bayesian non-linear regression

- Non-linearity in parameters with respect to the response variable, e.g. $y = a + x^b + \epsilon$; ($y = a + bx + cx^2 + \epsilon$ is a linear modeling)
- Parameters may have a specific biological meaning
- Non-linear models may be more parsimonious (i.e. fewer parameters) than a competitor linear model, such as a polynomial
- Bayesian non-linear modeling uses priors for all model parameters




UNIVERSITY OF GUELPH

# Ecotoxicological trial - earthworms exposed to ZnO:Mn




- Worms in contaminated soils (uptake) and clean soils (elimination)
- Different variances in Zn contents in the differerent phases

UNIVERSITY OF GUELPH

Van Lingen et al. (2025)

# Physiology-based toxicokinetic modeling

## Uptake phase: $0 < t \leq t_n$

$$C(t, t_n, C_{\exp}; \boldsymbol{\theta}) = C_0 + \frac{k_1}{k_2} C_{\exp}(1 - e^{-k_2 t}) \text{ with } \boldsymbol{\theta} = [C_0, k_1, k_2] \quad (1)$$

# Physiology-based toxicokinetic modeling

## Uptake phase: $0 < t \le t_n$

$$C(t, t_n, C_{\exp}; \boldsymbol{\theta}) = C_0 + \frac{k_1}{k_2} C_{\exp}(1 - e^{-k_2 t}) \text{ with } \boldsymbol{\theta} = [C_0, k_1, k_2] \quad (1)$$

## Elimination phase: $t > t_n$

$$C(t, t_n, C_{\exp}; \boldsymbol{\theta}) = C_0 + \frac{k_1}{k_2} C_{\exp}(e^{k_2(t - t_n)} - e^{-k_2 t}) \quad (2)$$

# Physiology-based toxicokinetic modeling

## Uptake phase: $0 < t \leq t_n$

$$C(t, t_n, C_{\text{exp}}; \boldsymbol{\theta}) = C_0 + \frac{k_1}{k_2} C_{\text{exp}}(1 - e^{-k_2 t}) \text{ with } \boldsymbol{\theta} = [C_0, k_1, k_2] \quad (1)$$

## Elimination phase: $t > t_n$

$$C(t, t_n, C_{\text{exp}}; \boldsymbol{\theta}) = C_0 + \frac{k_1}{k_2} C_{\text{exp}}(e^{k_2(t - t_n)} - e^{-k_2 t}) \quad (2)$$

## Equal or unequal variance along uptake and elimination phases?

$$y_{\text{ij}} = C(t, \ldots) + e_{\text{ij}} \text{ with } e_{\text{ij}} \sim N(0, \sigma_e^2) \quad (3)$$

$$y_{\text{ijk}} = C(t, \ldots) + e_{\text{ijk}} \text{ with } e_{\text{ijk}} \sim N(0, \delta_k \sigma_e^2) \quad (4)$$

Entity $i$ at time $j$ in phase $k$

UNIVERSITY OF GUELPH

# Physiology-based toxicokinetic modeling

## Uptake phase: $0 < t \leq t_n$

$$C(t, t_n, C_{\exp}; \boldsymbol{\theta}) = C_0 + \frac{k_1}{k_2} C_{\exp}(1 - e^{-k_2 t}) \text{ with } \boldsymbol{\theta} = [C_0, k_1, k_2] \quad (1)$$

## Elimination phase: $t > t_n$

$$C(t, t_n, C_{\exp}; \boldsymbol{\theta}) = C_0 + \frac{k_1}{k_2} C_{\exp}(e^{k_2(t - t_n)} - e^{-k_2 t}) \quad (2)$$

## Equal or unequal variance along uptake and elimination phases?

$$y_{\text{ij}} = C(t, \ldots) + e_{\text{ij}} \text{ with } e_{\text{ij}} \sim N(0, \sigma_e^2) \quad (3)$$

$$y_{\text{ijk}} = C(t, \ldots) + e_{\text{ijk}} \text{ with } e_{\text{ijk}} \sim N(0, \delta_k \sigma_e^2) \quad (4)$$

Entity $i$ at time $j$ in phase $k$

- $k_1$ and $k_2$ represent the uptake and elimination rates of the chemical

UNIVERSITY OF GUELPH

# Bayesian non-linear regression in STAN - 1

```
mod_NLin_ToxKin <- "
data {
  int<lower=0> N; // Number of observations
  real x[N];  // time
  real y[N];  // Zn concentration in earthworms
  real z; int t_d; // Zn concentration in soil and soil transfer time
  int grp[N]; int no_sigma; // Sigma stratification variable and number of sigma's

  real a_C0; real b_C0; // Prior parameters for C_0
  real mean_k1; real sd_k1; // Prior parameters for k_1
  real mean_k2; real sd_k2; // Prior parameters for k_2
} // End of data block

parameters {
  real<lower=0> C_0; real<lower=0> k[2]; // Contains C0, k1 and k2.
  real<lower=0> sigma_e[no_sigma]; // residual sd
} // End of parameters block

transformed parameters {
  real m[N];

  for (i in 1:N){
   m[i] = (x[i] <= t_d ? k[1]/k[2]*z*(1-exp(-k[2]*x[i])) :
                   k[1]/k[2]*z*(exp(-k[2]*(x[i]-t_d))-exp(-k[2]*x[i])));
  } // End of for loop i; note the meaning of '?' and ':'
} // End of transformed parameters block with one-compartmental function
 . . .
```

UNIVERSITY OF GUELPH

# Bayesian non-linear regression in STAN - 2

```
 . . .
model {
  // priors
  C_0 ~ normal(a_C0, b_C0); // Assign a gamma prior to C_0
  k[1] ~ normal(mean_k1, sd_k1); k[2] ~ normal(mean_k2, sd_k2); // Assign normal priors to k1 and k2
  sigma_e ~ cauchy(0,5); // Assign a half-cauchy prior to residual variance parameter sigma

  // likelihoods
  for (j in 1:N) y[j] ~ normal(C_0 + m[j], sigma_e[grp[j]]);   // Toxkin model
} // End of model block


generated quantities {
  // Initialize (additional) quantities to be generated
  vector[N] log_lik; // to use loo package and compute waic - the logLikelihood MUST be called log_lik!
  real pred[N]; // Quantities for fitted values (not including noise/error estimate)
  real res[N];  // Residuals

  // Assign values to the initialized quantities to be returned
  for (n in 1:N){
   pred[n] = C_0 + (x[n] <= t_d ? k[1]/k[2]*z*(1-exp(-k[2]*x[n])) :
                                        k[1]/k[2]*z*(exp(-k[2]*(x[n]-t_d))-exp(-k[2]*x[n])));
   res[n] = y[n] - pred[n];

   log_lik[n] = normal_lpdf(y[n] | pred[n], sigma_e[grp[n]]);   // Toxkin model
   } // End of for loop n

} // End of generated quantities block
"
```

# Bayesian non-linear regression in RSTAN - 1

```
library(rstan)
options(mc.cores = parallel::detectCores())
rstan_options(auto_write = TRUE)
library(loo)

### Compile stan model
stan_mod_NLin_ToxKin <- stan_model(model_code=mod_NLin_ToxKin, model_name="mod_NLin_ToxKin")

### Read and process data
worm_Zn_dat <- read.table("Efetida_Zn.txt", header=T, sep=",")
worm_Zn_dat <- data.frame(worm_Zn_dat, Phase=1+as.integer(worm_Zn_dat$Time > 14))

worm_dat.list <- list(N=nrow(worm_Zn_dat), # Number of observations
                      x=worm_Zn_dat$Time,  # time
                      y=worm_Zn_dat$toxicant,  # Zn concentration in earthworms
                      z=mean(worm_Zn_dat$C_exp), # Zn conc in soil
                      t_d=14, # soil transfer time
                      grp=rep(1,nrow(worm_Zn_dat)), # Sigma stratification variable
                      no_sigma=1, # number of sigma's
                      a_C0=75, b_C0=5, # Prior parameters for C_0
                      mean_k1=1, sd_k1=1, # Prior parameters for k_1
                      mean_k2=1, sd_k2=1) # Prior parameters for k_2

worm_Zn_homosc.rs <- sampling(stan_mod_NLin_ToxKin, data=worm_dat.list, iter=30e3,
         pars=c("C_0","k","sigma_e","res","log_lik"), warmup=1e3, thin=25, chains=2)

worm_dat.list$grp <- worm_dat.list$Phase
worm_dat.list$no_sigma <- 2
worm_Zn_heterosc.rs <- sampling(stan_mod_NLin_ToxKin, data=worm_dat.list, iter=30e3,
                          pars=c("C_0","k","sigma_e","res","log_lik"), warmup=1e3, thin=25, chains=2)
```

# Bayesian non-linear regression in RSTAN - 2

```r
print(summary(worm_Zn_homosc.rs, pars=c("C_0","k","sigma_e"))$summary)
```

```
##                mean se_mean    sd   2.5%    25%    50%    75%  97.5%   n_eff  Rhat
## C_0          73.645   0.170 4.794 64.431 70.171 73.578 76.815 83.035 795.424 1.000
## k[1]          0.530   0.005 0.129  0.318  0.435  0.517  0.608  0.812 715.261 1.000
## k[2]          2.405   0.022 0.596  1.471  1.975  2.323  2.761  3.736 719.660 0.999
## sigma_e[1] 44.371   0.215 5.801 34.646 40.204 43.923 47.880 57.740 727.920 0.999
```

```r
print(summary(worm_Zn_heterosc.rs, pars=c("C_0","k","sigma_e"))$summary)
```

```
##                mean se_mean     sd   2.5%    25%    50%    75%  97.5%    n_eff   Rhat
## C_0          70.806   0.034  1.832 67.130 69.639 70.827 72.030 74.326 2932.903 1.000
## k[1]          0.783   0.003  0.147  0.517  0.681  0.772  0.878  1.105 3231.801 0.999
## k[2]          3.614   0.009  0.519  2.712  3.241  3.570  3.938  4.756 3150.903 0.999
## sigma_e[1] 63.087   0.218 12.032 44.740 54.608 61.297 69.607 91.601 3039.504 1.000
## sigma_e[2]  7.370   0.027  1.474  5.145  6.309  7.161  8.196 10.846 2923.347 1.000
```

- Note the $\sigma_{e,1}$ and $\sigma_{e,2}$ values for the heteroscedastic model

UNIVERSITY OF GUELPH

# Bayesian non-linear prediction of Zn in earthworms




| Model fit | $C_0$ | $k_1$ | $k_2$ | $\sigma_e$ or $\sigma_{\mathrm{e,uptake}}$ | $\sigma_{\mathrm{e,elimin}}$ |
|---|---|---|---|---|---|
| Homoscedastic Zn | 73.6±4.8 | 0.530±0.129 | 2.41±0.60 | 44.4±5.8 | |
| Heteroscedastic Zn | 70.8±1.8 | 0.783±0.147 | 3.61±0.52 | 63.1±12.0 | 7.4±1.5 |

UNIVERSITY OF GUELPH

# Bayesian non-linear prediction of Zn in earthworms




| Model fit | $C_0$ | $k_1$ | $k_2$ | $\sigma_e$ or $\sigma_{\mathrm{e,uptake}}$ | $\sigma_{\mathrm{e,elimin}}$ |
|---|---|---|---|---|---|
| Homoscedastic Zn | 73.6±4.8 | 0.530±0.129 | 2.41±0.60 | 44.4±5.8 | |
| Heteroscedastic Zn | 70.8±1.8 | 0.783±0.147 | 3.61±0.52 | 63.1±12.0 | 7.4±1.5 |
| Homoscedastic Mn | 40.5±2.3 | 0.0515±0.0296 | 1.64±0.81 | 14.2±1.8 | |
| Heteroscedastic Mn | 40.6±2.0 | 0.0508±0.0306 | 1.67±0.80 | 18.0±3.4 | 9.62±1.81 |
| Homoscedastic Mn | 33.2±1.7 | 0.115±0.041 | 2.31±0.84 | 9.04±1.18 | |
| Heteroscedastic Mn | 33.2±1.6 | 0.114±0.042 | 2.27±0.85 | 10.1±1.9 | 8.11±1.51 |

# Bayesian non-linear residual variance modeling - residuals




UNIVERSITY OF GUELPH

# Bayesian non-linear regression in RSTAN - 3

```
compute_loo <- function(stan_out.rs){
  # Extract log likelihood; merge_chains=T indicates log-likelihoods merge across chains
  loglik_mod.rs <- extract_log_lik(stan_out.rs, parameter_name="log_lik", merge_chains=T)

  LL <- as.array(stan_out.rs, pars="log_lik" # Convert log Likelihood into an array

  # Compute rel. efficiency of the chains and then looic using log-likelihood and r_eff
  r_eff <- relative_eff(exp(LL), cores=1)
  ltest <- loo(loglik_mod.rs, r_eff=r_eff, cores=1)
  return(ltest) # Return LOOIC
} # End of compute_loo()

homosc.loo <- compute_loo(worm_Zn_homosc.rs)
heterosc.loo <- compute_loo(worm_Zn_heterosc.rs)

print(homosc.loo$estimates)
```

```
##            Estimate        SE
## elpd_loo -169.4749  6.719187
## looic     338.9498 13.438374
```

```
print(heterosc.loo$estimates)
```

```
##              Estimate         SE
## elpd_loo -145.834374  7.2614507
## looic     291.668747 14.5229014
```

- LOOIC obviously favors the heteroscedastic model

UNIVERSITY OF GUELPH

# Traceplots and density plots of model parameters




- Both plots visualize chain convergence, density plot visualizes the parameter distribution the best

# Posterior (predictive) distribution




- Predict future values of the response variable: $p(\tilde{y}|\boldsymbol{y}) = \int p(\tilde{y}|\theta)p(\theta|\boldsymbol{y})d\theta$
- Integrate whole parameter space out

UNIVERSITY OF GUELPH

# Posterior predictive distribution for Zn content in *E. fetida*




- Note the different credible interval widths in the uptake and elimination phases for the heteroscedastic model vs the homoscedastic model

# Posterior predictive distribution in STAN - 1

```
mod_NLin_ToxKin02 <- "

data {
  int<lower=0> N; // Number of observations
  real x[N];  // time
  real y[N];  // Zn concentration in earthworms
  real z; int t_d; // Zn concentration in soil and soil transfer time
  int grp[N]; int no_sigma; // Sigma stratification variable and number of sigma's
  int<lower=N> P; real pred_t[P]; int s_grp[P]; // ...

  real a_C0; real b_C0; // Prior parameters for C_0
  real mean_k1; real sd_k1; // Prior parameters for k_1
  real mean_k2; real sd_k2; // Prior parameters for k_2
} // End of data block

 . . .
```

- Posterior predictive distribution is simulated for P time points (`pred_t`) throughout uptake and elimination phases (`s_grp`)
- Parameter, transformed parameter and model blocks unchanged
- Code for Posterior predictive distribution in the generated quantities block...

# Posterior predictive distribution in STAN - 2

```
 . . .
generated quantities {
  // to use loo package and compute waic, the logLikelihood MUST be named log_lik!
  vector[N] log_lik;

  // Initialize (additional) quantities to be generated
  real pred[N]; // Quantities for fitted values (not including noise/error estimate)
  real res[N];  // Residuals
  real cpred[P]; real ypred[P]; // Predicted values (including noise/error estimate)

  // Assign values to the initialized quantities to be returned
  for (n in 1:N){
   pred[n] = C_0 + (x[n] <= t_d ? k[1]/k[2]*z*(1-exp(-k[2]*x[n])) :
                                k[1]/k[2]*z*(exp(-k[2]*(x[n]-t_d))-exp(-k[2]*x[n])));
   res[n] = y[n] - pred[n];

   log_lik[n] = normal_lpdf(y[n] | pred[n], sigma_e[grp[n]]); // Toxkin model

   } // End of for loop n

   for (p in 1:P){
   cpred[p] = (pred_t[p] <= t_d ? k[1]/k[2]*z*(1-exp(-k[2]*pred_t[p])) :
                         k[1]/k[2]*z*(exp(-k[2]*(pred_t[p]-t_d))-exp(-k[2]*pred_t[p])));

   ypred[p] = normal_rng(C_0 + cpred[p], sigma_e[s_grp[p]]); // Toxkin model

  } // End of for loop p
// End of generated quantities block
"
```

UNIVERSITY OF GUELPH

# Posterior predictive distribution RSTAN - 4

```
# Compile stan model
stan_mod_NLin_ToxKin02 <- stan_model(model_code=mod_NLin_ToxKin02, model_name="mod_NLin_ToxKin02")

# Data points for random number generator that samples posterior predictive distribution
t_rng <- seq(1,14,0.1)
rng_grp <- rep(1,length(t_rng)); t_rng <- c(t_rng,14+t_rng); rng_grp <- c(rng_grp,rng_grp+1)

# Generate data list object and run simulations
worm_dat.list02 <- list(N=nrow(worm_Zn_dat), #Number of observations
                        x=worm_Zn_dat$Time,  # time
                        y=worm_Zn_dat$toxicant,  # Zn concentration in earthworms
                        z=mean(worm_Zn_dat$C_exp), # Zn conc in soil
                        t_d=14, # soil transfer time
                        grp=rep(1,nrow(worm_Zn_dat)), # Sigma stratification variable
                        no_sigma=1, # number of sigma's
                        P=length(t_rng), pred_t=t_rng, s_grp=rng_grp,
                        a_C0=75, b_C0=5, # Prior parameters for C_0
                        mean_k1=1, sd_k1=1, # Prior parameters for k_1
                        mean_k2=1, sd_k2=1) # Prior parameters for k_2

worm_Zn_homosc.rs02 <- sampling(stan_mod_NLin_ToxKin02, data=worm_dat.list02, iter=30e3, warmup=1e3,
                                pars=c("C_0","k","sigma_e","log_lik","ypred"), thin=25, chains=2)

worm_dat.list02$grp<- worm_Zn_dat$Phase
worm_dat.list02$no_sigma <- 2
worm_Zn_heterosc.rs02 <- sampling(stan_mod_NLin_ToxKin02, data=worm_dat.list02, iter=30e3, warmup=1e3,
                                  pars=c("C_0","k","sigma_e","log_lik","ypred"), thin=25, chains=2)
```

- Generate timepoints from 1 to 14, then from 14 to 28; assign to `t_rng`

# Posterior predictive distribution RSTAN - 5

```r
print(summary(worm_Zn_homosc.rs02, pars=c("C_0","k","sigma_e","ypred"))$summary)
```

```
##                  mean se_mean     sd   2.5%     25%     50%     75%   97.5%    r_eff  Rhat
##C_0             73.633   0.052  4.582  64.629  70.543  73.625  76.662  82.565 7796.106 1.000
##k[1]             0.526   0.001  0.132   0.311   0.432   0.511   0.606   0.826 8004.950 1.000
##k[2]             2.382   0.007  0.599   1.408   1.939   2.315   2.753   3.714 7967.616 1.000
##sigma_e[1]      44.364   0.065  5.828  34.784  40.220  43.793  47.862  57.248 7942.351 1.000
##ypred[1]       207.577   0.532 46.710 114.044 177.534 207.569 238.773 297.928 7696.155 1.000
##ypred[2]       210.418   0.524 46.989 118.990 178.908 210.376 241.578 302.134 8045.789 1.000
```

. . .

- Note 2.5% and 97.5% percentiles of `ypred` quantifying the bounds of the gray shaded area in the plot

# *Gammarus pulex* toxicokinetic data example




Data from Miller et al. (2016) extracted using
`https://apps.automeris.io/wpd4/`

UNIVERSITY OF GUELPH

# *Gammarus pulex* **fits**




| Model fit | $C_0$ | $k_1$ | $k_2$ | $\sigma_e$ | $\delta$ |
|---|---|---|---|---|---|
| Homoscedastic | 0.922±0.648 | 0.563±0.054 | 0.0176±0.0029 | 3.718±0.513 | |
| Exponential | 0.527±0.325 | 0.461±0.043 | 0.0153±0.0023 | 0.758±0.258 | 0.111±0.024 |
| Power | 0.259±0.254 | 0.515±0.047 | 0.0159±0.0027 | 0.807±0.318 | 0.612±0.136 |

**Homoscedastic, exponential or power variance, respectively:**

$$e_{\mathrm{ij}} \sim N(0, \sigma^2), e_{\mathrm{ij}} \sim N(0, e^{2\delta y_{\mathrm{ij}}}\sigma^2) \text{ or } e_{\mathrm{ij}} \sim N(0, |y_{\mathrm{ij}}|^{2\delta}\sigma^2) \quad (5)$$

- Credible interval of $\delta$ does not contain zero $\rightarrow$ heteroscedasticity

# Residual plots of Bayesian non-linear regression




| | Homoscedastic | var Exp | var Pow |
|---|---|---|---|
| LOOIC | 167.6 | 139.7 | 149.6 |

- Lowest LOOIC for exponential residual variance model

# Posterior predictive distributions *G. pulex* models




- Exponential residual variance model follows data variance the best, but is not perfect with 2 points outside posterior predictive credible interval

# Thank you for your attention!




**Questions?**

# Exercise lecture 4

- Write STAN code for the *Gammarus pulex* data and use the various residual variance functions
- Insert as many `print()` statements as needed to check/understand what each line of code does
- Reconsider the priors and evaluate model simulation output